
Relativistic Spin Rotation

James J. Kelly

*Department of Physics
University of Maryland
College Park, MD 20742
jjkelly@physics.umd.edu*

It is well known that the spin direction of a particle is affected by a boost that is not parallel to its momentum, a relativistic effect described as *Wigner rotation* or *Wigner-Thomas precession*. Thus, we often require a transformation between polarizations expressed in either the lab or cm frames. Several versions of this transformation appear in the literature, but proving their equivalence can be algebraically tedious. Unfortunately, at least one widely cited paper that was intended to be pedagogical gets the transformation wrong. Therefore, in this notebook I review the standard derivation and then examine several of the published variations using *Mathematica* to perform as much of the tedious algebra as possible. Numerical results are provided for several typical reactions.

Created: September 13, 1996

Last revised: October 29, 2004

Initialization

```
ClearAll["Global`*"];
Off[General::spell, General::spell1];

$DefaultFont = {"Times", 12};
$TextStyle =
  {FontFamily -> "Times", FontSize -> 12, FontSlant -> "Italic"};

Needs["Miscellaneous`PhysicalConstants`"];
Needs["Miscellaneous`Units`"];
Needs["Graphics`Master`"];
Needs["Utilities`Notation`"];
Off[Convert::temp, Symbolize::boxSymbolExists]

Symbolize[ $\beta_1$ ]; Symbolize[ $\gamma_1$ ]; Symbolize[ $\theta_1$ ];
Symbolize[ $\beta_2$ ]; Symbolize[ $\gamma_2$ ]; Symbolize[ $\theta_2$ ]
```

```

MyAssumptions = { $\gamma > 1$ ,  $\gamma_1 > 1$ ,  $\gamma_2 > 1$ ,  $-1 < \beta < 1$ ,  $0 < \beta_1 < 1$ ,
   $0 < \beta_2 < 1$ ,  $0 < \theta_1 < \pi$ ,  $0 < \theta_2 < \pi$ ,  $\xi_1 > 0$ ,  $\xi_2 > 0$ ,  $\xi \in \text{Reals}$ };
MySimplify = Simplify[#, MyAssumptions] &;
MyFullSimplify = FullSimplify[#, MyAssumptions] &;

```

Wigner rotation

■ Derivation

The intrinsic spin of a massive particle is specified in its rest frame. Suppose that $\vec{\sigma}$ is a unit vector that represents the spin orientation in a rest frame. A spacelike spin four-vector is then formed as $s = (0, \vec{\sigma})$ such that $s_\mu s^\mu = -1$ is invariant. The spin four-vector in an arbitrary frame is obtained by applying the boost that represents the momentum in that frame. Nevertheless, the orientation of the spin is still represented by the original $\vec{\sigma}$. The relationship between spins in frames differing by a boost which is not parallel to the particle's momentum can be determined by transforming each frame to a rest frame of the particle, using a boost parallel to the momentum in that frame, and comparing the orientations of the two rest frames. These orientations differ because nonparallel boosts do not commute.

Consider two frames, S_1 and S_2 , that share a common set of coordinate axes and differ by a boost along \hat{z} . We assume, without loss of generality, that the velocity of a particle lies in the xz -plane and that the polar angles are in the range $0 \leq \theta_1, \theta_2 \leq \pi$. If we label the boost direction as \hat{z} and the normal to the plane containing the boost and momentum directions as \hat{y} , the four-vectors take the form $\{E, p_z, p_x\}$ where the y component is superfluous. The four-momenta can then be expressed as

$$\begin{aligned}
 p_1 &= m \gamma_1 \{1, \beta_1 \cos[\theta_1], \beta_1 \sin[\theta_1]\} \\
 p_2 &= m \gamma_2 \{1, \beta_2 \cos[\theta_2], \beta_2 \sin[\theta_2]\}
 \end{aligned}$$

in S_1 and S_2 , respectively. The momentum observed in frame S_2 moving with velocity $-\vec{\beta} = -\beta \hat{z}$ relative to S_1 is obtained via the boost $p_2 = B_z[-\beta].p_1$ where

$$B_z[\beta] = \begin{pmatrix} \gamma & -\gamma\beta & 0 \\ -\gamma\beta & \gamma & 0 \\ 0 & 0 & 1 \end{pmatrix}$$

Similarly, the momentum observed in a frame S_2 rotated by an angle θ relative to S_1 is given by $p_2 = R_y[\theta].p_1$ where

$$R_y[\theta] = \begin{pmatrix} 1 & 0 & 0 \\ 0 & \cos[\theta] & \sin[\theta] \\ 0 & -\sin[\theta] & \cos[\theta] \end{pmatrix}$$

Finally, if the relative velocity $\vec{\beta} = \beta_x \hat{x} + \beta_z \hat{z}$ lies at angle $\theta = \text{ArcTan}[\beta_z, \beta_x]$ in the xz -plane, we employ the combined boost

$$B[\theta, \beta] = R_y[-\theta].B_z[\beta].R_y[\theta]$$

Note that these transformation matrices are expressed in passive form, transforming the coordinate systems instead of the particle.

If the momentum forms an angle θ_1 with the \hat{z} axis of S_1 , we can reach a rest frame R_1 using a boost with velocity $\vec{\beta}_1$, such that

$$T_1 = B[\theta_1, \beta_1] = R_y[-\theta_1].B_z[\beta_1].R_y[\theta_1]$$

represents the complete Lorentz transformation from frame S_1 to a rest frame R_1 . Alternatively, we could first use a boost along the \hat{z} axis to transform from S_1 to S_2 in which the momentum p_2 appears at angle θ_2 . We then use a boost with velocity $\vec{\beta}_2$

$$T_2 = B[\theta_2, \beta_2].B_z[-\beta] = R_y[-\theta_2].B_z[\beta_2].R_y[\theta_2].B_z[-\beta]$$

to reach a rest frame R_2 . These are both rest frames of the same particle, so can differ at most by a rotation about their common \hat{y} axis, such that

$$T_2 = R_y[\chi].T_1$$

Thus, the angles between the spins in S_1 and S_2 can be obtained from the transformation

$$R_y[\chi] = T_2 T_1^{-1} = R_y[-\theta_2].B_z[\beta_2].R_y[\theta_2].B_z[-\beta].R_y[-\theta_1].B_z[-\beta_1].R_y[\theta_1]$$

Unfortunately, direct multiplication of these seven matrices produces a rather complicated result that is difficult to simplify and from which it is difficult to determine χ . A somewhat simpler method is to use that fact that successive rotations about the same axis add, such that

$$R_y[-\Omega] = B_z[\beta_2].R_y[\theta_2].B_z[-\beta].R_y[-\theta_1].B_z[-\beta_1]$$

where $\Omega = \theta_1 - \theta_2 - \chi$ is easier to evaluate. Once we demonstrate that this matrix represents a pure rotation, we can identify Ω from its matrix elements and deduce $\chi = \theta_1 - \theta_2 - \Omega$.

It is useful to distinguish between polarizations expressed with respect to a common coordinate system, described as z bases, from those expressed with respect to helicity bases with the longitudinal direction along the momentum for either frame. In z bases we find

$$\begin{pmatrix} P_z^{(2)} \\ P_x^{(2)} \end{pmatrix} = \begin{pmatrix} \cos[\chi] & \sin[\chi] \\ -\sin[\chi] & \cos[\chi] \end{pmatrix} \begin{pmatrix} P_z^{(1)} \\ P_x^{(1)} \end{pmatrix}$$

while for helicity bases we obtain

$$\begin{pmatrix} P_L^{(2)} \\ P_S^{(2)} \end{pmatrix} = \begin{pmatrix} \cos[\Omega] & -\sin[\Omega] \\ \sin[\Omega] & \cos[\Omega] \end{pmatrix} \begin{pmatrix} P_L^{(1)} \\ P_S^{(1)} \end{pmatrix}$$

where $\Omega = \theta_1 - \theta_2 - \chi$. The signs chosen for χ and Ω give positive angles when S_1 is the cm, S_2 is the lab, and $\beta > 0$. Notice that $\beta > 0 \implies \theta_2 < \theta_1$ corresponds to the typical cm→lab transformation while $\beta < 0 \implies \theta_2 > \theta_1$ corresponds to the typical lab→cm transformation. For small $|\beta|$ we expect χ to be small because it represents a relativistic effect that causes noncolinear Lorentz transformations not to commute while Galilean transformations would commute. Under these circumstances we expect to find $\Omega \approx \theta_1 - \theta_2$ to be positive for cm→lab transformations or negative for lab→cm transformations.

We are now ready to set up a *Mathematica* calculation of the spin rotation matrix. First we define the momenta in S_1 and S_2 as

$$\begin{aligned} \mathbf{p}_1 &= m \gamma_1 \{1, \beta_1 \cos[\theta_1], \beta_1 \sin[\theta_1]\}; \\ \mathbf{p}_2 &= m \gamma_2 \{1, \beta_2 \cos[\theta_2], \beta_2 \sin[\theta_2]\}; \end{aligned}$$

where $0 \leq \beta_1, \beta_2 \leq 1$. Next, we define the rotation matrix

$$\mathbf{R}_y[\theta_-] = \{\{1, 0, 0\}, \{0, \cos[\theta], \sin[\theta]\}, \{0, -\sin[\theta], \cos[\theta]\}\};$$

and boost functions

$$\begin{aligned} \mathbf{B}_z[\beta_-] &= \{\{\gamma, -\gamma\beta, 0\}, \{-\gamma\beta, \gamma, 0\}, \{0, 0, 1\}\} /. \{\gamma \rightarrow \frac{1}{\sqrt{1-\beta^2}}\}; \\ \mathbf{B}_z[\beta_-, \gamma_-] &= \{\{\gamma, -\gamma\beta, 0\}, \{-\gamma\beta, \gamma, 0\}, \{0, 0, 1\}\}; \end{aligned}$$

where, for algebraic reasons, it will prove convenient to define a variant in which the relationship between γ and β remains implicit. A boost in an arbitrary direction in the xz -plane is then

$$\begin{aligned} \mathbf{B}[\theta_-, \beta_-] &= \mathbf{R}_y[-\theta] \cdot \mathbf{B}_z[\beta] \cdot \mathbf{R}_y[\theta]; \\ \mathbf{B}[\theta_-, \beta_-, \gamma_-] &= \mathbf{R}_y[-\theta] \cdot \mathbf{B}_z[\beta, \gamma] \cdot \mathbf{R}_y[\theta]; \end{aligned}$$

Now we verify that a boost with $\vec{\beta} = \vec{\beta}_1$ transforms from S_1 to a rest frame R_1 such that

$$\begin{aligned} &\mathbf{B}[\theta_1, \beta_1] \cdot \mathbf{p}_1 /. \{\gamma_1 \rightarrow \frac{1}{\sqrt{1-\beta_1^2}}\} // \text{MySimplify} \\ &\{m, 0, 0\} \end{aligned}$$

Although the results are already familiar, we also check our functions by evaluating the relationship between p_2 and p_1 under a boost β in the z -direction.

$$\begin{aligned} &\text{Thread}[\mathbf{B}_z[-\beta, \gamma] \cdot \mathbf{p}_1 == \mathbf{p}_2] // \text{MySimplify} \\ &\{m \gamma \gamma_1 (1 + \beta \beta_1 \cos[\theta_1]) == m \gamma_2, \\ &\quad m \gamma \gamma_1 (\beta + \beta_1 \cos[\theta_1]) == m \beta_2 \gamma_2 \cos[\theta_2], m \beta_1 \gamma_1 \sin[\theta_1] == m \beta_2 \gamma_2 \sin[\theta_2]\} \end{aligned}$$

These results can be represented by substitution rules

$$\begin{aligned} \text{rule1} &= \{\gamma_2 \rightarrow \gamma_1 \gamma (1 + \beta \beta_1 \cos[\theta_1]), \sin[\theta_2] \rightarrow \frac{\beta_1 \gamma_1}{\beta_2 \gamma_2} \sin[\theta_1], \\ &\quad \cos[\theta_2] \rightarrow \frac{\gamma \gamma_1}{\beta_2 \gamma_2} (\beta_1 \cos[\theta_1] + \beta), \beta_2 \rightarrow \sqrt{1 - \frac{1}{\gamma_2^2}}\}; \\ \text{rule2} &= \{\gamma_1 \rightarrow \frac{1}{\sqrt{1-\beta_1^2}}, \gamma \rightarrow \frac{1}{\sqrt{1-\beta^2}}\}; \end{aligned}$$

We are now ready to evaluate the angle between R_1 and R_2 . Multiplying five transformations and applying the kinematic relationships between S_1 and S_2 gives

```

R12 =
(Bz[β2, γ2].Ry[θ2].Bz[-β, γ].Ry[-θ1].Bz[-β1, γ1] //. rule1) /. rule2 //
MyFullSimplify

```

$$\left\{ \{1, 0, 0\}, \left\{ 0, \frac{\beta_1 + \beta \cos[\theta_1]}{(1 + \beta \beta_1 \cos[\theta_1]) \sqrt{1 - \frac{(-1+\beta^2)(-1+\beta_1^2)}{(1+\beta \beta_1 \cos[\theta_1])^2}}}, \right. \right. \\
\left. \frac{\beta(-1+\beta_1^2) \sin[\theta_1]}{(1 + \beta \beta_1 \cos[\theta_1]) \sqrt{(1 - \beta_1^2) \left(1 - \frac{(-1+\beta^2)(-1+\beta_1^2)}{(1+\beta \beta_1 \cos[\theta_1])^2}\right)}} \right\}, \\
\left\{ 0, -\frac{\beta(-1+\beta_1^2) \sin[\theta_1]}{(1 + \beta \beta_1 \cos[\theta_1]) \sqrt{(1 - \beta_1^2) \left(1 - \frac{(-1+\beta^2)(-1+\beta_1^2)}{(1+\beta \beta_1 \cos[\theta_1])^2}\right)}} \right. \\
\left. \left. \frac{\beta_1 + \beta \cos[\theta_1]}{(1 + \beta \beta_1 \cos[\theta_1]) \sqrt{1 - \frac{(-1+\beta^2)(-1+\beta_1^2)}{(1+\beta \beta_1 \cos[\theta_1])^2}}} \right\} \right\}$$

where

```

R12.Transpose[R12] // MySimplify

```

```

{{1, 0, 0}, {0, 1, 0}, {0, 0, 1}}

```

```

Det[R12] // MySimplify

```

```

1

```

demonstrates that R_{12} is a proper orthogonal matrix or, in other words, a pure rotation. Thus, the net effect of three nonparallel coplanar boosts which take a system from one rest frame to another in the sequence $R_1 \rightarrow S_1 \rightarrow S_2 \rightarrow R_2$ is a pure rotation. Thus, by analyzing the components of this matrix

$$-\frac{R_{12}[[2, 3]]}{R_{12}[[2, 2]]} /. \{\beta^2 \rightarrow 1 - \frac{1}{\gamma^2}, \beta_1^2 \rightarrow 1 - \frac{1}{\gamma_1^2}, \beta_2^2 \rightarrow 1 - \frac{1}{\gamma_2^2}\} // \text{MySimplify} \\
\frac{\beta \sin[\theta_1]}{\beta_1 \gamma_1 + \beta \gamma_1 \cos[\theta_1]}$$

we identify

$$\Omega = \text{ArcTan}[\gamma_1 (\beta_1 + \beta \cos[\theta_1]), \beta \sin[\theta_1]] /. \\
\left\{ \gamma \rightarrow \frac{1}{\sqrt{1 - \beta^2}}, \gamma_1 \rightarrow \frac{1}{\sqrt{1 - \beta_1^2}} \right\} // \text{MyFullSimplify};$$

and use the quadrant-sensitive version of the inverse tangent. The rotation angle relative to a fixed set of coordinate axes is then

$$\chi = \theta_1 - \theta_2 - \Omega$$

or

$$\chi = \left(\theta_1 - \theta_2 - \Omega / . \theta_2 \rightarrow \text{ArcTan} \left[\frac{\beta_1 \text{Sin}[\theta_1]}{\gamma (\beta_1 \text{Cos}[\theta_1] + \beta)} \right] \right) / .$$

$$\left\{ \gamma \rightarrow \frac{1}{\sqrt{1 - \beta^2}}, \gamma_1 \rightarrow \frac{1}{\sqrt{1 - \beta_1^2}} \right\} / .$$

$$\text{ArcTan} \left[\frac{\gamma_-}{\gamma_+} \right] \rightarrow \text{ArcTan}[\mathbf{x}, \mathbf{y}] // \text{MyFullSimplify}$$

$$\theta_1 - \text{ArcTan} \left[\beta_1 + \beta \text{Cos}[\theta_1], \beta \sqrt{1 - \beta_1^2} \text{Sin}[\theta_1] \right] -$$

$$\text{ArcTan} \left[\beta + \beta_1 \text{Cos}[\theta_1], \sqrt{1 - \beta^2} \beta_1 \text{Sin}[\theta_1] \right]$$

Therefore, we can evaluate the spin rotation angle observed in the transformation $S_1 \rightarrow S_2$ produced by a boost $\vec{\beta} = \beta \hat{z}$ using

$$\chi = \theta_1 - \theta_2 - \Omega$$

$$\Omega = \text{ArcTan}[\gamma_1 (\beta_1 + \beta \text{Cos}[\theta_1]), \beta \text{Sin}[\theta_1]]$$

$$\theta_2 = \text{ArcTan}[\gamma (\beta_1 \text{Cos}[\theta_1] + \beta), \beta_1 \text{Sin}[\theta_1]]$$


These results are quite general and apply to any spin or to any pair of reference frames, although the most common applications are probably to lab \leftrightarrow cm transformations for spin one-half. Also note that β may be positive or negative.

The following function returns the kinematics and spin rotation angles in the form of replacement rules. We will use this below to display some specific examples.

```
transform[ $\beta 1\_$ ,  $\theta 1\_$ ,  $\beta\_$ ] := Module[{ $\gamma 1$ ,  $\gamma$ ,  $\Omega$ ,  $\theta 2$ ,  $\chi$ ,  $\gamma 2$ },
 $\gamma 1 = \frac{1}{\sqrt{1 - \beta 1^2}}$ ;  $\gamma = \frac{1}{\sqrt{1 - \beta^2}}$ ;
 $\theta 2 = \text{ArcTan}[\gamma (\beta 1 \text{Cos}[\theta 1] + \beta), \beta 1 \text{Sin}[\theta 1]]$ ;
 $\Omega = \text{ArcTan}[\gamma 1 (\beta 1 + \beta \text{Cos}[\theta 1]), \beta \text{Sin}[\theta 1]]$ ;
 $\chi = \theta 1 - \theta 2 - \Omega$ ;
 $\gamma 2 = \gamma 1 \gamma (1 + \beta \beta 1 \text{Cos}[\theta 1])$ ;
{theta2  $\rightarrow \theta 2$  / Degree, chi  $\rightarrow \chi$  / Degree, omega  $\rightarrow \Omega$  / Degree, gamma2  $\rightarrow \gamma 2$ }]
```

Suppose that S_1 is the cm frame while S_2 is the lab frame reached by a boost with $\beta > 0$. The momentum then appears to rotate forward, such that $\theta_2 \leq \theta_1$. We also find, at least numerically, that the spin rotates in same direction but by a much smaller angle.

```
Plot[Evaluate[{theta2, chi} /. transform[0.2,  $\theta_1$  Degree, 0.5]],
      { $\theta_1$ , 0, 180},
      PlotRange -> {Automatic, All},
      PlotStyle -> {RGBColor[0, 0, 1], RGBColor[1, 0, 0]},
      Frame -> True, FrameLabel -> {" $\theta_1$ ", ""},
      PlotLegend -> {" $\theta_2$ ", " $\chi$ "}, LegendPosition -> {0., 0},
      LegendSize -> {0.6, 0.3}, LegendShadow -> None];
```



■ example: nucleon-nucleon scattering

An important special case occurs when the projectile scatters from a particle of equal mass, such that

$$m_1 = m_2 \implies \beta = \beta_c = \frac{\beta_\ell}{2} \implies \Omega = \theta_\ell \quad \chi = \theta_c - 2\theta_\ell$$

where β is the boost from the cm to the lab frame and where S_1 and S_2 are now labeled as S_c and S_ℓ , respectively. Substituting these kinematics, we find

$$\theta_\ell = \Omega = \text{ArcTan}[\gamma_c (1 + \text{Cos}[\theta_c]), \text{Sin}[\theta_c]]$$

such that

$$\chi = \theta_c - 2\theta_\ell$$

Finally, in the nonrelativistic limit, we find that


$$\gamma \rightarrow 1 \implies \theta_c = 2\theta_\ell \implies \chi \rightarrow 0$$

justifies the identification of χ as the "relativistic" spin rotation angle. More generally, $\theta_c > 2\theta_\ell$ for equal masses so that $\chi > 0$. Therefore, we conclude that successive nonparallel boosts rotate the spin in the same direction but generally much less than the momentum.

The figure below shows χ as a function of cm scattering angle for laboratory kinetic energies ranging from 0.25 to 2.25 GeV in steps of 0.5 GeV, with the smallest (largest) energy producing the smallest (largest)

Wigner rotation. The maximum Wigner rotation angle under these condition is only about 8° . It is a small effect, but still must be included in the analysis of depolarization or polarization transfer data.

```
Plot[
  Evaluate[Table[chi /. transform[ $\frac{\beta}{2}$ ,  $\theta_1$  Degree,  $\frac{\beta}{2}$ ] /. { $\beta \rightarrow \frac{\sqrt{\gamma^2 - 1}}{\gamma}$ } /.
     $\gamma \rightarrow \frac{T + 1}{0.939}$ , {T, 0.25, 2.25, 0.5}]], { $\theta_1$ , 0, 180}, PlotPoints -> 50,
  Frame -> True, FrameLabel -> {" $\theta_c$  (deg)", " $\chi$  (deg)"},
  PlotLabel -> "Wigner Rotation for NN Scattering",
  Epilog -> Text[" $T_{lab}=\{0.25, 2.25, 0.5\}$  GeV", {90, 0.5}]];
```



■ example: pion electroproduction on the nucleon

Let S_1 represent the cm frame and S_2 the lab frame for pion electroproduction on the nucleon $eN \rightarrow eN\pi$. The following functions evaluate the lab angle and Wigner precession angle in terms of the cm angle given $\{W, Q^2\}$. The example below is relevant to JLab E91-011, but larger precession angles would be seen for more extreme kinematics (large W or large Q^2).

```
eNπRules = { $\gamma \rightarrow \frac{m_p^2 + Q^2 + W^2}{2 m_p W}$ ,  $\gamma_1 \rightarrow \frac{m_p^2 + W^2 - m_\pi^2}{2 m_p W}$ ,  $m_p \rightarrow 0.939$ ,  $m_\pi \rightarrow 0.14$ };


eNπ[Win_, Q2_, θ1_] :=
  transform[ $\frac{\sqrt{\gamma_1^2 - 1}}{\gamma_1}$ , θ1 Degree,  $\frac{\sqrt{\gamma^2 - 1}}{\gamma}$ ] /. eNπRules /.
  { $Q^2 \rightarrow Q2$ ,  $W \rightarrow Win$ }
```



```


Plot[Evaluate[{theta2, chi} /. eNpi[1.232, 1.0, theta1]],
  {theta1, 0, 180}, PlotRange -> {Automatic, All}, PlotPoints -> 50,
  PlotStyle -> {RGBColor[0, 0, 1], RGBColor[1, 0, 0]},
  Frame -> True, FrameLabel -> {"theta_c", " "},
  PlotLabel -> "eN -> eNpi for (W, Q^2) = (1.232, 1.0)",
  PlotLegend -> {"theta_l", "chi"}, LegendPosition -> {0., -0.3},
  LegendSize -> {0.6, 0.2}, LegendShadow -> None];

```



Notice that lab spin is slightly forward of the cm spin while the lab momentum is considerably forward. However, recoil polarization is usually represented in a helicity basis with the longitudinal component along the momentum either in the cm or the lab, as appropriate. Thus, the helicity basis for this reaction rotates forward by the angle $\theta_c - \theta_l$ in transforming from the cm to the lab frame. This is usually a large angle that produces considerable mixing between the helicity states. The apparent spin rotation relative to the helicity bases for the two frames is then determined by $\Omega = \theta_c - \theta_l - \chi$, which the figure below shows is generally large and negative. Thus, a spin which appears to be purely longitudinal in the cm frame appears to be transverse in the lab when θ_c is near about 105° for these kinematics.

```
Plot[Evaluate[{omega} /. eNpi[1.232, 1.0,  $\theta_1$ ]],
      { $\theta_1$ , 0, 180}, PlotRange -> {Automatic, All},
      PlotPoints -> 50, Frame -> True, FrameLabel -> {" $\theta_c$ ", " $\Omega$ "},
      PlotLabel -> "eN→eNπ for (W,Q²)=(1.232,1.0)";
```



■ derivation using rapidity

Several authors, including both Dmitrasinovic and Giebink, perform their analyses using the rapidity representation. In the hope that this is algebraically simpler, we now modify our derivation along these lines. First, we define the rapidity boost, evaluate the matrix product, and test its orthogonality.

```
boost[ξ_] = {{Cosh[ξ], -Sinh[ξ], 0}, {-Sinh[ξ], Cosh[ξ], 0}, {0, 0, 1}};
```

```
p1 = m {Cosh[ξ1], Sinh[ξ1] Cos[θ1], Sinh[ξ1] Sin[θ1]};
```

```
p2 = m {Cosh[ξ2], Sinh[ξ2] Cos[θ2], Sinh[ξ2] Sin[θ2]};
```

```
Thread[boost[-ξ].p1 == p2] // MySimplify
```

```
{m Cosh[ξ] Cosh[ξ1] + m Cos[θ1] Sinh[ξ] Sinh[ξ1] == m Cosh[ξ2],
 m Cosh[ξ1] Sinh[ξ] + m Cos[θ1] Cosh[ξ] Sinh[ξ1] == m Cos[θ2] Sinh[ξ2],
 m Sin[θ1] Sinh[ξ1] == m Sin[θ2] Sinh[ξ2]}
```

```
rule3 = {ξ2 -> ArcCosh[Cosh[ξ1] Cosh[ξ] + Sinh[ξ] Sinh[ξ1] Cos[θ1]],
```

```
  Sin[θ2] ->  $\frac{\text{Sinh}[\xi_1]}{\text{Sinh}[\xi_2]} \text{Sin}[\theta_1],$ 
```

```
  Cos[θ2] ->  $\frac{1}{\text{Sinh}[\xi_2]} (\text{Sinh}[\xi_1] \text{Cosh}[\xi] \text{Cos}[\theta_1] + \text{Cosh}[\xi_1] \text{Sinh}[\xi])$ };
```

```

test = boost[ξ2].Ry[θ2].boost[-ξ].Ry[-θ1].boost[-ξ1] // . rule3 //
MyFullSimplify

{ {1, 0, 0}, {0, (Cos[θ1] Cosh[ξ1] Sinh[ξ] + Cosh[ξ] Sinh[ξ1]) /
  (1 + Cosh[ξ] Cosh[ξ1] + Cos[θ1] Sinh[ξ] Sinh[ξ1])
  √{1 - 2 / (1 + Cosh[ξ] Cosh[ξ1] + Cos[θ1] Sinh[ξ] Sinh[ξ1])} },
  - (Sin[θ1] Sinh[ξ]) / (1 + Cosh[ξ] Cosh[ξ1] + Cos[θ1] Sinh[ξ] Sinh[ξ1])
  √{1 - 2 / (1 + Cosh[ξ] Cosh[ξ1] + Cos[θ1] Sinh[ξ] Sinh[ξ1])} },
  {0, (Sin[θ1] Sinh[ξ]) / (1 + Cosh[ξ] Cosh[ξ1] + Cos[θ1] Sinh[ξ] Sinh[ξ1])
  √{1 - 2 / (1 + Cosh[ξ] Cosh[ξ1] + Cos[θ1] Sinh[ξ] Sinh[ξ1])} },
  (Cos[θ1] Cosh[ξ1] Sinh[ξ] + Cosh[ξ] Sinh[ξ1]) /
  (1 + Cosh[ξ] Cosh[ξ1] + Cos[θ1] Sinh[ξ] Sinh[ξ1])
  √{1 - 2 / (1 + Cosh[ξ] Cosh[ξ1] + Cos[θ1] Sinh[ξ] Sinh[ξ1])} } }

test.Transpose[test] // MySimplify

{{1, 0, 0}, {0, 1, 0}, {0, 0, 1}}

Det[test] // MySimplify

1

```

Unfortunately, this result is not appreciably simpler than the previous version. Nevertheless,

$$\begin{aligned}
& - \frac{\text{test}[[2, 3]]}{\text{test}[[2, 2]]} // \text{MySimplify} \\
& \frac{\text{Sin}[\theta_1] \text{Sinh}[\xi]}{\text{Cos}[\theta_1] \text{Cosh}[\xi_1] \text{Sinh}[\xi] + \text{Cosh}[\xi] \text{Sinh}[\xi_1]}
\end{aligned}$$

agrees with the previous $\text{Tan}[\Omega]$.

Alternative representations

Many different versions of these formulas appear in the literature, but it is often difficult to demonstrate their equivalence symbolically. Our formulation has the advantage that the formulas are relatively simple, much simpler than others that appear in the literature, but has the disadvantage that χ must be evaluated in two steps — we compute θ_2 and Ω separately and then combine them to obtain $\chi = \theta_1 - \theta_2 - \Omega$ — and it is not immediately obvious that χ is small for most reactions. In the remainder of this section, we compare our results with other representations, symbolically when possible or numerically when not.

■ Comparison with Dmitrasinovic

Dmitrasinovic obtained the following formula using a somewhat different derivation in Phys. Rev. C 47, 2195 (1993). Here we translate his Eq. (21) into the present notation.

$$\omega = 2 \operatorname{ArcTan} \left[\frac{\gamma \beta \gamma_1 \beta_1 \sin[\theta_1]}{(1 + \gamma)(1 + \gamma_1) + \gamma \beta \gamma_1 \beta_1 \cos[\theta_1]} \right];$$

It is difficult to prove that these two approaches give identical results, but after many attempts I was able to find a successful approach using *Mathematica*. Note that because *Mathematica* does not simplify expressions involving the two-argument form of ArcTan very well, it was necessary to replace those functions with the single-argument form. I have not yet checked whether this replacement has consequences for some argument pairs.

$$\left(\operatorname{Tan}[\chi - \omega] /. \left\{ \gamma \rightarrow \frac{1}{\sqrt{1 - \beta^2}}, \gamma_1 \rightarrow \frac{1}{\sqrt{1 - \beta_1^2}} \right\} /. \right. \\ \left. \operatorname{ArcTan}[x_, y_] \rightarrow \operatorname{ArcTan}\left[\frac{y}{x}\right] // \operatorname{TrigExpand} \right) // \operatorname{MyFullSimplify}$$

0

I have also verified numerically, using many choices of β and β_1 , that these formulas agree to machine precision over the entire angular range when the two-argument arc tangent is retained. To ensure that possible quadrant problems are tested, we include both signs for β and values of $|\beta|$ both larger and smaller than β_1 . Therefore, even though the proof is not completely rigorous, I am confident that his ω is mathematically identical to our χ .


$$\mathbf{diff} = \chi - \left(\omega /. \left\{ \gamma \rightarrow \frac{1}{\sqrt{1-\beta^2}}, \gamma_1 \rightarrow \frac{1}{\sqrt{1-\beta_1^2}} \right\} \right)$$

$$\Theta_1 - 2 \operatorname{ArcTan} \left[\frac{\beta \beta_1 \sin[\Theta_1]}{\sqrt{1-\beta^2} \sqrt{1-\beta_1^2} \left(\left(1 + \frac{1}{\sqrt{1-\beta^2}} \right) \left(1 + \frac{1}{\sqrt{1-\beta_1^2}} \right) + \frac{\beta \beta_1 \cos[\Theta_1]}{\sqrt{1-\beta^2} \sqrt{1-\beta_1^2}} \right)} \right] -$$


$$\operatorname{ArcTan}[\beta_1 + \beta \cos[\Theta_1], \beta \sqrt{1-\beta_1^2} \sin[\Theta_1]] -$$

$$\operatorname{ArcTan}[\beta + \beta_1 \cos[\Theta_1], \sqrt{1-\beta^2} \beta_1 \sin[\Theta_1]]$$


```
Plot[Evaluate[diff /. {β1 → 0.6, β → 0.3}], {Θ1, 0, π}];
```




```
Plot[Evaluate[diff /. {β1 → 0.6, β → 0.8}], {Θ1, 0, π}];
```




```
Plot[Evaluate[diff /. { $\beta_1 \rightarrow 0.6$ ,  $\beta \rightarrow -0.3$ }], { $\theta_1$ , 0,  $\pi$ }] ;
```



```
Plot[Evaluate[diff /. { $\beta_1 \rightarrow 0.6$ ,  $\beta \rightarrow -0.8$ }], { $\theta_1$ , 0,  $\pi$ }] ;
```



```
Plot[Evaluate[diff /. { $\beta_1 \rightarrow 0.6$ ,  $\beta \rightarrow 0.6$ }], { $\theta_1$ , 0,  $\pi$ }] ;
```



However, the discussion given by Dmitrasinovic has caused considerable confusion in several analyses of recent JLab experiments on recoil polarization. According to this paper, "The angle ω is the precession angle of the spin with respect to the coordinate system defined by the direction of motion of the particle, and will be referred to as the Wick-Wigner helicity precession angle." In an earlier paragraph he claims "... ω is the *precession* angle with respect to the *rotated* direction of motion". His figures show that ω is small, which suggests that if the spin is purely longitudinal in one frame it should be nearly longitudinal in another even if the momentum directions are quite different. His Eq. (26) suggests that laboratory response functions are obtained from cm response functions using a rotation through the angle ω , which produces relatively little mixing between longitudinal and transverse components. Later when examining the nonrelativistic limit, he claims "... there is no spin rotation relative to the momentum, i.e. no Wick-Wigner precession, in the nonrelativistic limit: The spin and the velocity three-vectors remain parallel in that limit". Near the end of the paper he states "... the spin direction essentially rotates together with the nucleon momentum" in the deuteron electrodissintegration reaction with close to nonrelativistic kinematics. This interpretation strongly conflicts with the present analysis in which χ is the precession angle with respect to a fixed coordinate system with the longitudinal direction along the relative velocity between the two frames. The precession angle with respect to the helicity representation is then given by Ω which would correspond to the angle $-\gamma$ in his notation. Perhaps I am still misunderstanding his words, but they seem clear enough. So it comes down to the question of whether spin is affected nonrelativistically by a Galilean transformation.

In standard nonrelativistic quantum mechanics we express the wave function for a particle in the factorized form

$$\Psi_\alpha = \psi[\vec{p}, \vec{r}] \Phi_\alpha$$

where $\psi[\vec{p}, \vec{r}]$ governs its motion and Φ_α represents its internal degrees of freedom, such as its spin orientation. A Galilean transformation affects ψ but does not affect Φ if there is no rotation of the coordinate system or change of the spin quantization axis. Under those conditions the spin direction remains fixed relative to a fixed coordinate system, but if the momentum changes direction the helicity will also. There seems to be no basis for the claim that the spin direction rotates with the velocity direction under a Galilean transformation. There is even an interpretation in Dmitrasinovic's paper, attributed to Sommerfeld, that the relativistic spin rotation arises because velocities and momenta add differently, resulting in a hyperbolic velocity triangle that does not comply with Euclidean geometry. The precession angle can then be described as a *hyperbolic defect* because those angles do not sum to 180° .

From a purely classical point of view we would define spin as the internal angular momentum of a system of masses about their center of mass, whereby

$$\vec{s} = \sum_i (\vec{r}_i - \vec{R}) \times (\vec{p}_i - \vec{P})$$

where

$$M = \sum_i m_i, \quad \vec{R} = \frac{1}{M} \sum_i m_i \vec{r}_i, \quad \vec{P} = \sum_i m_i \vec{v}_i$$

Under a Galilean transformation

$$\vec{v}_i \longrightarrow \vec{v}_i + \vec{v}_0, \quad \vec{P} \longrightarrow \vec{P} + M \vec{v}_0 \implies \vec{s} \longrightarrow \vec{s}$$

we immediately find that the spin direction is preserved. Granted quantum mechanical spin for an elementary particle cannot be reduced to the sum of relative orbital angular momenta for its constituents, but the relativistic spin rotation is an essentially classical effect that does not depend upon quantum mechanics at all. Therefore, the claim that the spin direction follows the velocity direction violates the correspondence principle.

Fortunately, I believe that these misconceptions were always remedied prior to publication and that no published data have been adversely affected. Nevertheless, a lot of time has been wasted by initially accepting the results and discussion in Dmitrasinovic's paper at face value.

■ Giebink

Although Giebink Phys. Rev. C 32, 502 (1985) is primarily concerned with the construction and transformation properties of few-body states, he provides one of the simplest derivations of the spin transformation and gives the same formula for $\text{Tan}[\Omega]$ that is derived here. He also states unequivocally, that "in the nonrelativistic limit, there is no spin precession in the z basis" and provides a formula for the mixing of helicity basis states using a rotation through the angle Ω . Finally, he provides a formula for $\text{Sin}[\Omega]$

$$\text{Sin}[\Omega] = \frac{1 + \gamma}{\gamma_1 + \gamma_2} \text{Sin}[\theta_1 - \theta_2] = \frac{1 + \text{Cosh}[\xi]}{\text{Cosh}[\xi_1] + \text{Cosh}[\xi_2]} \text{Sin}[\theta_1 - \theta_2]$$

in which the nonrelativistic limit $\Omega \rightarrow \theta_1 - \theta_2$ is obvious. However, in using this formula to find Ω one must take care to account for situations in which $\text{Cos}[\Omega] < 0$. From

$$\text{Tan}[\Omega] = \frac{\beta \text{Sin}[\theta_1]}{\gamma_1 (\beta_1 + \beta \text{Cos}[\theta_1])} = \frac{\text{Sinh}[\xi] \text{Sin}[\theta_1]}{\text{Cosh}[\xi] \text{Sinh}[\xi_1] + \text{Sinh}[\xi] \text{Cosh}[\xi_1] \text{Cos}[\theta_1]}$$

it is clear that quadrant ambiguities occur when $\beta > \beta_1$ permits $\beta_1 + \beta \text{Cos}[\theta_1]$ to change sign. Recognizing that $\text{Sin}[\theta_1]$ is nonnegative for $0 \leq \theta_1 \leq \pi$, the following calculation

$$\begin{aligned} \text{giebink} &= \frac{1 + \gamma}{\gamma_1 + \gamma_2} \text{Sin}[\theta_1 - \theta_2]; \\ \text{temp} &= (\text{giebink} // \text{TrigExpand}) // . \text{rule1} /. \\ &\quad \left\{ \gamma \rightarrow \frac{1}{\sqrt{1 - \beta^2}}, \gamma_1 \rightarrow \frac{1}{\sqrt{1 - \beta_1^2}} \right\} // \text{MyFullSimplify}; \\ \text{Tan}[\text{ArcSin}[\text{temp}]] & // \text{MySimplify} \\ &= \frac{\beta \sqrt{1 - \beta_1^2} \text{Sin}[\theta_1]}{\text{Abs}[\beta_1 + \beta \text{Cos}[\theta_1]]} \end{aligned}$$

demonstrates that this result has the same magnitude but lacks the quadrant sensitivity of the formula for $\text{Tan}[\Omega]$.

Also note that the variation given in Appendix B of Arenhövel, Leideman, and Tomusiak, Few-Body Systems 15, 109 (1993) is based upon the results of Giebink. They do not mention the possible quadrant ambiguity, but I hope that it is accounted for in their calculations.

■ Wijesooriya

Wijesooriya *et al.*, Phys. Rev. C66, 034614 (2002) provide an appendix with essentially the same derivation used here (multiplication of five matrices) but do not simplify the results. Instead, they identify $\text{Sin}[\Omega]$ and $\text{Cos}[\Omega]$ from appropriate matrix elements without substituting the kinematics of p_2 .

```
test2 = boost[ξ2].Ry[θ2].boost[ξ].Ry[-θ1].boost[-ξ1] // MyFullSimplify

{-Sinh[ξ1] (Cos[θ1] Cosh[ξ2] Sinh[ξ] +
  (Cos[θ1] Cos[θ2] Cosh[ξ] + Sin[θ1] Sin[θ2]) Sinh[ξ2]) +
  Cosh[ξ1] (Cosh[ξ] Cosh[ξ2] + Cos[θ2] Sinh[ξ] Sinh[ξ2]),
-Cosh[ξ1] (Cos[θ1] Cosh[ξ2] Sinh[ξ] +
  (Cos[θ1] Cos[θ2] Cosh[ξ] + Sin[θ1] Sin[θ2]) Sinh[ξ2]) +
  Sinh[ξ1] (Cosh[ξ] Cosh[ξ2] + Cos[θ2] Sinh[ξ] Sinh[ξ2]),
Cosh[ξ2] Sin[θ1] Sinh[ξ] +
  (Cos[θ2] Cosh[ξ] Sin[θ1] - Cos[θ1] Sin[θ2]) Sinh[ξ2]},
{-Cosh[ξ1] (Cos[θ2] Cosh[ξ2] Sinh[ξ] + Cosh[ξ] Sinh[ξ2]) +
  Sinh[ξ1] (Cosh[ξ2] (Cos[θ1] Cos[θ2] Cosh[ξ] + Sin[θ1] Sin[θ2]) +
    Cos[θ1] Sinh[ξ] Sinh[ξ2]),
-Sinh[ξ1] (Cos[θ2] Cosh[ξ2] Sinh[ξ] + Cosh[ξ] Sinh[ξ2]) +
  Cosh[ξ1] (Cosh[ξ2] (Cos[θ1] Cos[θ2] Cosh[ξ] + Sin[θ1] Sin[θ2]) +
    Cos[θ1] Sinh[ξ] Sinh[ξ2]), Cos[θ1] Cosh[ξ2] Sin[θ2] -
  Sin[θ1] (Cos[θ2] Cosh[ξ] Cosh[ξ2] + Sinh[ξ] Sinh[ξ2])},
{Cosh[ξ1] Sin[θ2] Sinh[ξ] +
  (Cos[θ2] Sin[θ1] - Cos[θ1] Cosh[ξ] Sin[θ2]) Sinh[ξ1],
Cosh[ξ1] (Cos[θ2] Sin[θ1] - Cos[θ1] Cosh[ξ] Sin[θ2]) +
  Sin[θ2] Sinh[ξ] Sinh[ξ1],
Cos[θ1] Cos[θ2] + Cosh[ξ] Sin[θ1] Sin[θ2]}

{test2[[2, 3]], test2[[2, 2]]} //.
{Cosh[ξ] → γ, Sinh[ξ] → β Cosh[ξ], Cosh[ξ1] → γ1, Sinh[ξ1] → β1 Cosh[ξ1],
  Cosh[ξ2] → γ2, Sinh[ξ2] → β2 Cosh[ξ2]} // MyFullSimplify

{-γ γ2 (β β2 + Cos[θ2]) Sin[θ1] + γ2 Cos[θ1] Sin[θ2],
  γ1 γ2 (γ Cos[θ1] (β β2 + Cos[θ2]) - β1 γ (β2 + β Cos[θ2]) + Sin[θ1] Sin[θ2])}
```

These results are consistent with those given by Wijesooriya *et al.*